

    

        struct_inspect

        v0.1.4



    


  

    Table of contents

    
      



      	StructInspect


      	Usage Guide


      	LICENSE





        	
          Modules
          


      	StructInspect


      	StructInspect.Opts


      	StructInspect.Overrides





        



      

    

  

    StructInspect

[image: CI]
[image: Hex.pm]
[image: Conventional Commits]
[image: Documentation]
[image: License: MIT]
StructInspect is a configurable library to customize struct and map inspection in Elixir. It allows you to omit fields with "empty" values, producing cleaner and more readable output.
For more details and advanced usage and configuration examples, please read the Usage Guide.
Installation
The package can be installed by adding struct_inspect to your list of dependencies in mix.exs:
def deps do
  [
    {:struct_inspect, "~> 0.1.4"}
  ]
end
Usage
To enable StructInspect for a particular struct, you just use StructInspect in your module definition.
defmodule MyStruct do
  use StructInspect

  defstruct [:id, :name, :bio, :email]
end
By default, StructInspect will omit fields with nil values, empty strings, empty lists, and empty structs.
iex> struct = %MyStruct{id: 1, name: "Gemini", bio: nil, email: ""}
%MyStruct{id: 1, name: "Gemini"}
Configuration
The behavior of StructInspect is determined by a set of options that can be provided at different levels. The core of the configuration is the StructInspect.Opts module, which defines all the available options and their default values. Please refer to the StructInspect.Opts module documentation for the most up-to-date list of defaults.
You can customize the inspection behavior on a per-struct basis by passing options to the use StructInspect macro. You can pass a map, a keyword list, or a list of atoms.
For example, if you want to also omit fields with false values, you can do:
defmodule MyConfiguredStruct do
  use StructInspect, false_value: true

  defstruct [:id, :name, :is_active]
end
iex> struct = %MyConfiguredStruct{id: 1, name: "Gemini", is_active: false}
%MyConfiguredStruct{id: 1, name: "Gemini"}
When a list of atoms is provided, ONLY the named keys will be set to true and the rest to false, overriding all the defaults.
defmodule MyOtherStruct do
  use StructInspect, [:false_value, :true_value]

  defstruct [:id, :name, :is_active, :is_legacy]
end
Application-wide Configuration
You can also configure the default options for StructInspect application-wide in your config/config.exs file.
# This setting will be merged to the defaults in `StructInspect.Opts`.
config :struct_inspect,
  ommits: [
    nil_value: true,
    empty_string: true,
    false_value: true
  ]
# This setting will produce a `StructInspect.Opts` set with only nils, empty maps and zero integers enabled.
config :struct_inspect,
  ommits: [:nil_value, :empty_map, :zero_integer]
The options passed to use StructInspect will be merged with the application-wide configuration. The options passed directly to use have precedence.
Available Options
The following options are available in StructInspect.Opts:
	  nil_value: Omits nil values. (default: true)
	  zero_integer_value: Omits 0. (default: false)
	  zero_float_value: Omits 0.0. (default: false)
	  empty_string: Omits "". (default: true)
	  empty_list: Omits []. (default: true)
	  empty_map: Omits %{}. (default: true)
	  empty_struct: Omits empty structs. (default: true)
	  empty_tuple: Omits {}. (default: true)
	  true_value: Omits true. (default: false)
	  false_value: Omits false. (default: false)
	  except: List of keys to omits. (default: [:__struct__])

License
StructInspect is released under the MIT License. See the LICENSE file for more details.


  

    Usage Guide

StructInspect provides a configurable implementation of the Inspect protocol for Elixir structs. It allows developers to define rules for omitting fields with specific values—such as nil, "", or []—from the inspection output. This can result in a more concise representation of data structures, especially in logging and interactive sessions.
Getting Started
To begin using StructInspect, add it to your list of dependencies in your mix.exs file:
def deps do
  [
    {:struct_inspect, "~> 0.1.4"}
  ]
end
Then, run mix deps.get to fetch the new dependency.
Basic Usage
The simplest way to use StructInspect is to use it in your struct definition.
defmodule MyStruct do
  use StructInspect

  defstruct [:id, :name, :bio, :email]
end
By default, StructInspect will omit fields with nil values, empty strings (""), empty lists ([]), empty tuples ({}), empty structs (%SomeStruct{}) and key :struct.
Let's see it in action:
iex> struct = %MyStruct{id: 1, name: "Gemini", bio: nil, email: ""}
%MyStruct{id: 1, name: "Gemini"}
As you can see, the :bio and :email fields are not present in the inspected output because their values are nil and "" respectively.
Deriving vs. Using StructInspect
In Elixir, you can use @derive Inspect to provide an inspect implementation for your structs.
defmodule User do
  @derive {Inspect, only: [:name]}
  defstruct [:name, :email]
end
The @derive Inspect macro allows for customization through the :only, :except, and :optional options. The :optional option is similar to what StructInspect does, but it requires you to explicitly list all the fields that should be considered optional. For more information, see the official documentation.
StructInspect provides an alternative. Instead of specifying which fields to omit, you define rules based on the values of the fields. By using use StructInspect, you can customize the inspect output to omit fields with certain values, such as nil, empty strings, or empty lists. This helps create a more concise representation of your data structures, especially in logging and interactive sessions, without having to list every possible optional field.
In summary:
	  @derive Inspect: Provides an inspect implementation with customization options (:only, :except, :optional). You need to specify the fields to be considered optional.
	  use StructInspect: Provides a configurable inspect implementation that allows you to omit fields based on their values.

Use StructInspect when you need to control the inspect output of your structs by defining omission rules based on values rather than field names.
Configuration
StructInspect offers a flexible configuration system that allows you to tailor the inspection behavior to your specific needs. The configuration is managed by the StructInspect.Opts module, which defines all the available options and their default values.
You can configure StructInspect in three ways:
	 Per-Struct Configuration: Directly in your struct module.
	 Application-wide Configuration: Globally in your config/config.exs file.
	 Combination: A mix of both, with per-struct configuration taking precedence.

Per-Struct Configuration
You can customize the inspection behavior for a specific struct by passing options to the use StructInspect macro. You can provide these options as a map, a keyword list, or a list of atoms.
Using a Keyword List
This is the most common way to configure a struct. You can override the default options by providing a keyword list.
For example, to also omit fields with false values:
defmodule MyConfiguredStruct do
  use StructInspect, false_value: true

  defstruct [:id, :name, :is_active]
end
Now, when a MyConfiguredStruct is inspected, fields with a value of false will be omitted:
iex> struct = %MyConfiguredStruct{id: 1, name: "Gemini", is_active: false}
%MyConfiguredStruct{id: 1, name: "Gemini"}
Using a List of Atoms
When you provide a list of atoms, only the specified keys will be set to true, and all other options will be set to false, completely overriding the defaults.
defmodule MyOtherStruct do
  use StructInspect, [:false_value, :true_value]

  defstruct [:id, :name, :is_active, :is_legacy]
end
In this case, only fields with false or true values will be omitted. All other "empty" values like nil or "" will be shown.
Application-wide Configuration
For consistent inspection behavior across your entire application, you can set default options in your config/config.exs file.
# In your config/config.exs

# This setting will be merged with the defaults in `StructInspect.Opts`.
config :struct_inspect,
  ommits: [
    nil_value: true,
    empty_string: true,
    false_value: true
  ]
You can also provide a list of atoms to enable only specific options globally:
# This setting will produce a `StructInspect.Opts` set with only nils, empty maps and zero integers enabled, and the except with [:__struct__].
config :struct_inspect,
  ommits: [:nil_value, :empty_map, :zero_integer]
The options passed directly to use StructInspect in a module will be merged with the application-wide configuration, with the per-struct options taking precedence.
Available Omissions
Here is a complete list of all the available omission options that you can use to configure StructInspect.
	  nil_value (default: true): Omits fields with a value of nil.
	  zero_integer_value (default: false): Omits fields with an integer value of 0.
	  zero_float_value (default: false): Omits fields with a float value of 0.0.
	  empty_string (default: true): Omits fields with an empty string value ("").
	  empty_list (default: true): Omits fields with an empty list ([]).
	  empty_map (default: true): Omits fields with an empty map (%{}).
	  empty_struct (default: true): Omits fields that contain an "empty" struct. See 'What is an "Empty Struct"?' below.
	  empty_tuple (default: true): Omits fields with an empty tuple ({}).
	  true_value (default: false): Omits fields with a boolean value of true.
	  false_value (default: false): Omits fields with a boolean value of false.
	  except (default: [:__struct__]): Hides the listed fields from the struct or map.

Note on except: The except option is for hiding fields by their name, while the other omission options hide fields based on their value. Fields listed in except will always be hidden.

What is an "Empty Struct"?
A struct is considered "empty" if all of its values are the same as a newly created struct without setting any keys. In other words, if struct == %SomeStruct{}.
For example, consider this struct:
defmodule Address do
  use StructInspect

  defstruct [:street, :city, zip_code: "00000-0000", po_box?: false]
end
If you create an Address struct like this, it is considered empty, because :street, :city are nil and the fields :zip_code and :po_box? contain theirs default values, therefore:
If you have a User struct that contains a address struct without changes:
defmodule User do
  use StructInspect

  defstruct [:name, :address]
end

user = %User{name: "Gemini", address: %Address{}}
When you inspect the user struct, the :address field will be omitted because the address struct itself is considered empty.
iex> user
%User{name: "Gemini"}
This helps to keep your logs and console output clean, especially when dealing with nested structs.
Overriding Structs from Dependencies
StructInspect provides a clean way to customize the inspection of structs from your dependencies without altering their source code. This can be done by defining overrides in your configuration.
Let's take, for example, the Phoenix.LiveView.Socket struct, which is a notoriously large struct.
Phoenix.LiveView.Socket with default Inspect protocol implementation:
iex(1)> %Phoenix.LiveView.Socket{}
#Phoenix.LiveView.Socket<
  id: nil,
  endpoint: nil,
  view: nil,
  parent_pid: nil,
  root_pid: nil,
  router: nil,
  assigns: %{__changed__: %{}},
  transport_pid: nil,
  sticky?: false,
  ...
>
Configure the Overrides
In your config/config.exs file, add the :overrides configuration for struct_inspect. You can list the modules you want to override.
# in config/config.exs
config :struct_inspect,
  overrides: [
    Phoenix.LiveView.Socket
  ]
Now with the override in place, here is the same output for Phoenix.LiveView.Socket struct:
iex(1)> %Phoenix.LiveView.Socket{}
%Phoenix.LiveView.Socket{
  private: %{live_temp: %{}},
  assigns: %{__changed__: %{}},
  sticky?: false
}
By default, the configuration, will use the standard StructInspect options. If you want to specify, what type of contents are to be omitted, for a particular struct, you can use a tuple where the second element can be a keyword, list of atoms or a map, see the Configuration section in this document for options configuration, for example:
# in config/config.exs or test/config.exs
config :struct_inspect,
  overrides: [
    {Phoenix.LiveView.Socket, [nil_value: false, empty_struct: false]},
    Another.Module,
    {SpecialStruct, [:nil_value, :empty_string]}
  ]
In the example above, for Phoenix.Live_View.Socket, we are specifying that we are overriding the defaults by allowing nil values and empty struct to be outputted. Another.Module will take the StructInspect.Opts defaults, and SpecialStruct will only omit nil values and empty strings.
Enabling the Overrides
To enable the configured overrides, you need to use StructInspect.Overrides in a file that is compiled by the Elixir compiler.
For the test environment, a good place for this is in your test/test_helper.exs file:
# in test/test_helper.exs
ExUnit.start()

use StructInspect.Overrides
For the dev and prod environments, you can create a new file, for example lib/struct_inspect_overrides.ex, with the following content:
# in lib/struct_inspect_overrides.ex
defmodule MyApp.StructInspectOverrides do
  use StructInspect.Overrides
end
This will ensure that the Inspect protocol is overridden for the configured modules when your application is compiled.
Handling Compiler Warnings
When you use the :overrides configuration, the Elixir compiler will issue a warning for each module you are overriding, similar to this:
warning: redefining protocol implementation for Inspect for Phoenix.LiveView.Socket
This is expected. You are intentionally replacing the default Inspect implementation for that module with a custom one provided by StructInspect.
It is recommended to keep these warnings visible to be aware of the protocol overrides. However, once you have acknowledged them, you can suppress them by setting the ignore_compiler_warning option to true in your config/config.exs:
# in config/config.exs
config :struct_inspect,
  ignore_compiler_warning: true
This will set the ignore_module_conflict and ignore_already_consolidated compiler options, effectively hiding the warnings for the protocol overrides.
Overriding the Map Module
StructInspect allows you to override the Inspect protocol for the Map module. This is a feature that can help you to control how maps are inspected in your application.
Warning: Overriding the Inspect protocol for the Map module is a significant change that can affect the behavior of your application and its dependencies in unexpected ways. It is recommended to use this feature only in the test environment.
To override the Map module, you need to add it to the :overrides configuration in your config/test.exs file:
# in config/test.exs
config :struct_inspect,
  overrides: [
    Map
  ]
This will cause all maps to be inspected using the StructInspect rules, which can be useful for cleaner test output.
Examples
Here are a few examples demonstrating how to combine different configuration options.
Example 1: Omitting nil and false values
Let's define a struct where we want to omit fields that are nil or false.
defmodule UserPreferences do
  use StructInspect, [:nil_value, :false_value]

  defstruct [:user_id, :notifications_enabled, :theme, :show_hidden_files]
end

# With default `nil_value: true` and `false_value: true`
iex> prefs = %UserPreferences{user_id: 123, notifications_enabled: true, theme: nil, show_hidden_files: false}
%UserPreferences{user_id: 123, notifications_enabled: true}
In this case, :theme is omitted because it's nil, and :show_hidden_files is omitted because it's false.
Example 2: Using except to hide a specific field
You can use the except option to hide a field by its name, regardless of its value. For instance, you might want to hide a sensitive token.
defmodule Session do
  use StructInspect, except: [:__struct__, :token]

  defstruct [:session_id, :user_id, :token]
end

iex> session = %Session{session_id: "abc-123", user_id: 42, token: "super-secret-token"}
%Session{session_id: "abc-123", user_id: 42}
Here, the :token field is never shown in the inspected output.
Example 3: Combining value-based omissions and except
You can combine value-based omissions with the except option.
defmodule Report do
  # Since the given options are a keyword list, they are combined with the 
  # StructInspect.Opts default state, so empty list, empty string, empty maps 
  # and empty structs are also ommitted
  use StructInspect, false_value: true, except: [:__struct__, :raw_data]

  defstruct [:id, :name, :details, :processed, :raw_data]
end

# :processed is false, so it's omitted.
# :raw_data is in the except list, so it's omitted.
# :details is omitted because it contains a empty string value.
iex> report = %Report{id: 1, name: "Sales", details: "", processed: false, raw_data: %{some: "data"}}
%Report{id: 1, name: "Sales"}

# Even if :processed is true, :raw_data is still omitted.
iex> report = %Report{id: 2, name: "Inventory", processed: true, raw_data: %{some: "data"}}
%Report{id: 2, name: "Inventory", processed: true}

MIT License

Copyright (c) 2025 Federico Alcántara Tatis
Copyright (c) 2025 WAdvanced

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.

Provides a mechanism to customize struct inspection by omitting fields with empty values.
By use-ing StructInspect in your struct definition, it automatically implements the
Inspect protocol for you, providing a more compact and readable representation of your
structs, especially when they contain many optional or empty fields.
Key Features
	Automatic Inspect Implementation: Simplifies the process of custom struct inspection.
	Configurable Empty Values: Allows customization of what is considered an "empty"
value through application configuration.
	Clean and Readable Output: Produces a cleaner inspection output by hiding noise from
empty fields.
	Empty Structs: A struct is considered empty if all its fields have the default
values defined in its defstruct. This is particularly useful for nested structs that
may be initialized with default values but are not considered "empty" by default.

Examples
First, you need to define a struct and use StructInspect:
defmodule MyStruct do
  @enforce_keys [:id]
  defstruct [:id, :name, :age, :data]
  use StructInspect
end

# Now, when you inspect instances of `MyStruct`, fields with empty values will be omitted:

%MyStruct{id: 1, name: "John", age: nil, data: %{}}
> %MyStruct{id: 1, name: "John"}
You can also configure the default omitted values in your config/config.exs:
config :struct_inspect, :ommits, [:nil_value, :empty_string]

        Generates the Inspect protocol implementation for the struct.



    


    
      
        __using__(opts \\ [])

      


        Injects the necessary code to enable custom inspection for a struct.



    


    
      
        inspect(map, opts, ommits)

      


        Filters the struct fields and builds the inspection algebra.



    


    
      
        inspect(map, name, infos, opts)

      


        Builds the inspection algebra for a map.

          @spec __before_compile__(Macro.Env.t()) :: Macro.t()


      


Generates the Inspect protocol implementation for the struct.
This macro is called before the module is compiled and defines the Inspect
implementation. The implementation delegates the actual inspection logic to the
inspect/4 function.

  



    

  
    
      
    
    
      __using__(opts \\ [])


        (macro)


        
          
        

    

  


  

      

          @spec __using__(keyword()) :: Macro.t()


      


Injects the necessary code to enable custom inspection for a struct.
This macro sets up a @before_compile hook to generate the Inspect protocol
implementation for the calling module. It also reads the default omitted values from the
application environment.

          @spec inspect(
  map() | struct(),
  Inspect.Opts.t(),
  [atom()] | keyword() | StructInspect.Opts.t()
) :: Inspect.Algebra.t()


      


Filters the struct fields and builds the inspection algebra.
Parameters
	data (map() | struct()) - The struct or map to be inspected.

	opts (Inspect.Opts.t()) - The inspection options.
	ommits (keyword()) - A list of atoms representing the types of empty values to omit.

Returns
Inspect.Algebra.t() - The algebra document representing the compacted struct.

          @spec inspect(map(), binary(), list(), Inspect.Opts.t()) :: Inspect.Algebra.t()


      


Builds the inspection algebra for a map.
Parameters
	map (map()) - The map to be inspected.
	name (binary()) - The name of the map.
	infos (list()) - A list of fields to be inspected.
	opts (Inspect.Opts.t()) - The inspection options.

Returns
Inspect.Algebra.t() - The algebra document representing the compacted map.

Provides a struct to configure the behavior of StructInspect.

        apply_to_defaults(attrs \\ [])

      


        Applies the given attributes to the default options.



    


    
      
        change(options, ommits)

      


        Changes an existing StructInspect.Opts struct with new ommits.



    


    
      
        new(ommits \\ [])

      


        Creates a new StructInspect.Opts struct.

          @type attributes() :: t() | map() | keyword() | [atom()] | atom()


      



  



  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %StructInspect.Opts{
  empty_list: boolean(),
  empty_map: boolean(),
  empty_string: boolean(),
  empty_struct: boolean(),
  empty_tuple: boolean(),
  except: list(),
  false_value: boolean(),
  nil_value: boolean(),
  true_value: boolean(),
  zero_float_value: boolean(),
  zero_integer_value: boolean()
}

          @spec apply_to_defaults(attributes()) :: t()


      


Applies the given attributes to the default options.
The default options are taken from the application environment.
Parameters
	attrs (attributes()) - The attributes to apply. Defaults to [].

Returns
t() - A new StructInspect.Opts struct with the applied attributes.

          @spec change(t(), attributes()) :: t()


      


Changes an existing StructInspect.Opts struct with new ommits.
Parameters
	options (t()) - The options struct to change.
	ommits (attributes()) - The new ommits to apply.

Returns
t() - The modified StructInspect.Opts struct.

          @spec new(attributes()) :: t()


      


Creates a new StructInspect.Opts struct.
It can receive:
	A map or keyword list of ommits to override the default values.
	A list of atoms, where only the named keys are set to true and the rest to false.
	Another StructInspect.Opts struct, which is returned as is.
	Nothing, in which case it returns the default struct.

Parameters
	ommits (attributes()) - The ommits to create the struct. Defaults to nil.

Returns
t() - A new StructInspect.Opts struct.

Provides a mechanism to globally override the Inspect implementation for structs.
This module allows you to configure StructInspect to take over the inspection of any
struct, even for libraries or dependencies where you cannot add use StructInspect
directly.
The overrides are configured in your config/config.exs file.

        __using__(opts)

      


        Generates Inspect implementations for the modules configured in the application environment.

      __using__(opts)


        (macro)


        
          
        

    

  


  

      

          @spec __using__(keyword()) :: Macro.t()


      


Generates Inspect implementations for the modules configured in the application environment.
This macro reads the :struct_inspect, :overrides configuration and for each module,
it generates a defimpl Inspect that uses StructInspect.inspect/3 for inspection.

